
ITG SUBJECT: CIRCULAR TEMPERATURE RECORDING CHART MEASUREMENTS

INTRODUCTION

Circular temperature recording charts are used in a broad variety of firms from food canneries to medical
device manufacturers to drug manufacturers. Investigators currently use draftman's dividers to measure
chord (straight line, point-to-point) instead of arc (curve) distances to check process time. There is no
easy way to measure arcs in investigational work. Firms also use dividers or stamped metal gauges to do
the same chord measurements. When an investigator wants to set dividers to a 25 minute chord, for
instance, he/she normally checks the setting by spanning 2 (on charts with 10 minute markings) or 3 (on
charts with 15 minute markings) estimated 25 minute chords to come up with an even marked time (50 and
75 minutes on 10 and 15 minute marked charts, respectively). He/she repeats the process until successful.
Once the dividers come up on the even markings, the investigator assumes the dividers are properly set.
However, estimating odd times such as 13, 19, or 23 minutes is difficult using this method.

In addition to the previous problem, a one minute time difference is only about 0.014 inch on a typical
Taylor OP524 chart. Introducing a 0.014 inch error by making 15 spans or misstamping a gauge is easy to
envision.

Another problem is the extreme difficulty in extrapolating times from known chord lengths. An investigator
cannot calculate a 17 minute chord distance accurately, given either longer (20, 30 minutes) or shorter
(10, 15 minutes) chord distances, because of the simple fact that two chords representing shorter times
(i.e., 10 minutes each) have a combined length longer than a 20 minute chord.

These chord problems are compounded when a firm uses different process temperatures, handles multiple
products and/or containers, or uses different temperature recording charts.

COMPUTER-CALCULATED CHORDS

The geometry for determining the arc or chord length of a circle is shown in Appendix 1. A computer can be
used to generate reference tables of chord lengths related to times for any circular recording chart at any
temperature. An investigator only needs to know: (1) the number of hours on the chart, and (2) the
diameter of the circle for the particular temperature desired. Geometric theory is based on the fact that a
radius bisecting a chord creates two equal right triangles.

APPLICATION

It is suggested that personnel modify the attached computer programs which were written in Microsoft
BASIC. Program Listing 1, written for a Tandy 200 laptop portable computer, is compatible on most
microcomputers. Program Listing 2 was written for a TRS-80 Model 1 computer. Both programs should run
on IBM-PCs unchanged (except possibly for adding spaces where required by GWBASIC). The programs can
be adapted to any district computer by knowledgeable personnel. Test the resultant programs by using
values from a 24 hour Taylor OP600 chart. The diameter of a circle for 240 degrees F is 7.929 inches.
Results should be 0.017299 for 1 minute, 0.034596 for 2 minutes, 0.051895 for 3 minutes, 0.172970 for 10
minutes, and 2.052180 for 120 minutes. Program Listing 2 can be used to print reference tables while
Program Listing 1 is primarily a demonstration program. The most useful application of the programs would
be to develop reference tables and maintain them in EI files. Once a blank chart is collected and expected

process temperatures are run through Program Listing 2, the resultant tables (similar to Table 1 attached)
can be kept in EI files and photocopies can be taken to inspection sites. Another possible application is to
take laptop computers to inspection sites. A computer used in conjunction with a dot matrix printer can
also generate chord gauges on paper similar to the metal gauges made in cannery machine shops.

Ouestions about the geometric theory or computer programming can be addressed to DEIO/IB, (301)443-
3340.

CIRCLE DIAGRAM 4

Appendix 1. Geometric Theory (Diagram not available)

1. An isosceles triangle is formed by connecting a chord (C) with the center of a circle. Call (A) the
angle in the center of the circle.

2. A line through the center of the circle, perpendicular to the chord, bisects the chord and its arc.

3. Since the bisecting line is perpendicular to the chord, each half of the original triangle is a right
triangle.

4. Each right triangle formed above has a small angle equal to A/2 Let A/2 = a.

5. Therefore Sin a = C/2 divided by the radius (R) or, Sin a = C/2R = C/D where D is the diameter of the
circle.

6. Solving for the chord, C = D Sin a. The chord represents the time which the investigator measures
with dividers.

7. We want to make chord values equal to minutes. It is simplest to standardize on the fact that there
are 360 degrees in a circle. If we assign one minute equal one degree to angle A and its chords, there
are 720 angle a's (degrees) in the circle. Thus, we create a formula where a (minutes) = a (degrees)
divided by 2, or more apprpriately t (minutes) = a (degrees)/2.

8. We must also consider a factor to correct different chart values to minutes because while there are
360 minutes on a 6 hour chart, there are 720 minutes on a 12 hour chart and 1440 minutes on a 24
hour chart. From these figures, factor F = 6/H where H is the number of hours on the chart.

9. Combining the above two formulas, t (minutes) = a (degrees) 1/2. Placing this in step 6 above, we get
chord = diameter Sin (3t/H) or C = D Sin (3t/H). This is the formula used in Program Listings 1 and 2.

Program Listing 1

1 REM DARRELL LEE, SALT LAKE CITY RESIDENT POST FTS 588-5285
10 DIM C(360): K = .0174533: CLS : REM DIMENSIONS VARIABLES,
INSERTS RADIAN CORRECTION FACTOR (K) 
100 INPUT "Enter chart name and number"; CH$ 
110 INPUT "Enter no. hours on chart"; H 
120 F = H / 3: REM F IS CORRECTION FACTOR 
130 INPUT "Enter target temperature for chord lengths"; TE 
140 PRINT "Enter diameter of"; TE; : INPUT " degree circle
    in inches or millimeters"; D 
150 IF D > 15 THEN M$ = "millimeter" ELSE M$ = "inch" 
160 INPUT "Is this correct (y/n)"; A$ 
170 IF LEFT$(A$, 1) = "y" OR LEFT$(A$, 1) = "Y" THEN 200 ELSE 100 
200 PRINT "Table of times with corresponding chord lengths
    for a(n) "; CH$; 
    " chart at "; TE; " degrees (a"; D; M$; " diameter circle)." 
210 PRINT : PRINT "Minutes "; 
220 IF M$ = "millimeter" THEN S$ = "s" ELSE S$ = "es" 
230 PRINT M$; S$: PRINT 

240 FOR T = 1 TO 120 STEP 4 
250 FOR TT = T TO T + 3 
260 C(TT) = D * SIN(K * TT / F) 
270 NEXT TT 
280 PRINT T; C(T): PRINT T + 1; C(T + 1): PRINT T + 2;
    C(T + 2): PRINT T + 3; C(T + 3) 
290 NEXT T

Program Listing 2

 

10 ' REPORT CHART PLANNER 
20 ' RUNS ON RADIO SHACK TRS-80 COMPUTER. MODIFY FOR OTHER COMPUTERS 
NOT USING MICRSOFT BASIC. 
30 CLEAR 2000 35 LPRINT CHR$(3): REM PRINTER COMMAND RESETS PAGE 
40 DIM C(360): K = .0174533: 
   A$ = "### ###.###### ### ###.###### ### ###.
   ###### ### ###.###### ### ###.######" 
50 ' K = RADIAN CORRECTION FACTOR. NOT USED IF COMPUTER CALUCULATES 
SINE USING DEGREES. 
60 CLS 
70 FOR I = 1 TO 8: PRINT : NEXT I 
80 PRINT "********** RETORT CHART PLANNER          **********" 
90 FOR I = 1 TO 500: NEXT: FOR I = 1 TO 6: PRINT : NEXT I 
100 PRINT "           ***** DARRELL LEE                  ****" 
110 PRINT "           * 2073 E. GYRFALCON DR.             *" 
120 PRINT "           * SANDY. UT 84092                   *" 
130 PRINT "           * SALT LAKE CITY RESIDENT POST      *" 
140 PRINT "           * ROOM 1033 ADMINISTRATION BLDG.    *" 
150 PRINT "           * 1745 WEST 1700 SOUTH              *" 
160 PRINT "           * SALT LAKE CITY. UT 84104          *" 
170 PRINT "           ************************************" 
180 PRINT : PRINT 
190 FOR I = 1 TO 1000: NEXT I 
200 INPUT "ENTER RETORT CHART MANUFACTURER'S NUMBER
    (IE. TAYLOR OP 524)"; O$ 
210 INPUT "ENTER NUMBER OF HOURS ON RETORT CHART (6,12,24)"; H 
220 F = H / 3 
260 INPUT "ENTER TARGET TEMPERATURE FOR CHORD LENGTHS"; TE 
270 PRINT "ENTER DIAMETER OF"; TE; : INPUT "DEGREE CIRCLE 
    IN INCHES OR MILLIMETERS"; D 
280 IF D > 15 THEN M$ = "MM. ": GOTO 300 
290 M$ = "INCH " 300 LPRINT "TABLE OF TIMES WITH CORRESPONDING 
    CHORD LENGTHS FOR A(N) ": LPRINT 
310 LPRINT O$:' CHART AT";TE; "DEGREES (A";D;M$;"DIAMETER CIRCLE)":LPRINT 
320 LPRINT "FOR ADDITIONAL CUSTOM MADE DATA TABLES CONTACT 
DARRELL LEE AT SALT LAKE CITY RESIDENT POST.": LPRINT : LPRINT : LPRINT 
330 FOR T = 1 TO 120 STEP 4 
340 FOR TT = T TO T + 3 
350 C(TT) = D * SIN(K * TT / F): NEXT TT 
360 LPRINT USINGGA$; T, C(T), T + 1, C(T + 1), 
    T + 2, C(T + 2), T +3, C(T + 3) 
370 NEXT T 
380 INPUT "DO YOU WISH TO PRINT ANOTHER TABLE"; Q$ 
390 IF LEFT$(Q$, 1) = "Y" THEN 30 ELSE END

 

Table 1. Chart Example

TABLE OF TIMES WITH CORRESPONDING CHORD LENGTHS FOR A(N) TAYLOR OP600 CHART AT 240 DEGREES
(A 7.929 INCH DIAMETER CIRCLE) FOR ADDITIONAL CUSTOM MADE DATA TABLES CONTACT DARRELL LEE AT

SALT LAKE CITY RESIDENT POST

Time   Chord(in) Time   Chord(in)    Time  Chord(in)  Time    Chord(in) 
(min)            (min)               (min)            (min) 
____  _________   _____  _________   ___   _________ _______   ___________ 
 1    0.017299      2    0.034596     3     0.051895    4        0.069193 
 5    0.086491      6    0.103787     7     0.121084    8        0.138580 
 9    0.155675     10    0.172970    11     0.190264   12        0.207557 
13    0.224849     14    0.242140    15     0.259429   16        0.276718 
17    0.294006     18    0.311292    19     0.328575   20        0.345858 
21    0.363140     22    0.380418    23     0.397696   24        0.414972 
25    0.432245     26    0.449517    27     0.466787   28        0.484054 
29    0.501319     30    0.518582    31     0.535842   32        0.553099 
33    0.570354     34    0.587607    35     0.604856   36        0.622102 
37    0.639346     38    0.656587    39     0.673824   40        0.691058 
41    0.708289     42    0.725516    43     0.742741   44        0.759961 
45    0.777178     46    0.794391    47     0.811601   48        0.828807 
49    0.846008     50    0.863206    51     0.880400   52        0.897588 
53    0.914774     54    0.931955    55     0.949130   56        0.966302 
57    0.983470     58    1.000630    59     1.017790   60        1.034940 
61    1.052090     62    1.069230    63     1.086370   64        1.103500 
65    1.120630     66    1.137750    67     1.154870   68        1.171980 
69    1.189090     70    1.206190    71     1.223280   72        1.240370 
73    1.257450     74    1.274530    75     1.291600   76        1.308660
77    1.325720     78    1.342770    79     1.359820   80        1.366860 
81    1.393890     82    1.410910    83     1.417930   84        1.444950 
85    1.461950     86    1.478950    87     1.49540    88        1.512930
89    1.529900     90    1.546870    91     1.563830   92        1.580790 
93    1.597740     94    1.614680    95     1.631610   96        1.648530 
97    1.665450     98    1.682360    99     1.699260  100        1.716150
101   1.733040    102    1.749910   103     1.766780  104        1.783640
105   1.800490    106    1.817330   107     1.834160  108        1.850990 
109   1.867810    110    1.884610   111     1.901410  112        1.918200 
113   1.934980    114    1.951750   115     1.968510  116        1.985260
117    2.002010   118    2.018740   119     2.035460  120        2.052180

 

NOTE: Darrell Lee is now a National Expert in Computers and is located in the SAN-DO office.

